WS1.5

Coming home …

Imagine that you come home tired after a long day at work. Your car has found the way to your house all by itself and is now turned into your street.
It drives to your property, stops and turns so it can drive to its pitch. The gate to your property opens, your car rolls up to its pitch and stops there. Now the lights are on the way to your front door, so you're safe to come to the house. You press the bell button. In the hall the light goes on and someone opens and greets you. Then you go to different rooms. Near the doors there are switches and everyone has the possibility to switch off or on the light in the hall from any switch.

You need *two* programs:

1. Rover

Construction:

Fix an external ranger front right (fig. 1).

Use a 30cm long cable and connect it to IN1.



fig. 1

Getting started, Rover

- drives slowly forward and onboard-LED is white
- scans the right side of the street using the external ranger
- stops (LED is red) and drives a curve to the right (LED is yellow and blinking)
- drives again slowly forward (LED is white) and stops before the gate using the built-in ranger, LED is red
- waits until the gate is open
- drives forward (LED is white) and stops in front of the wall (built-in ranger, LED is red)



fig. 2



fig. 3



fig. 4



fig. 5



fig. 6

2. House (innovator)

Fig. 2 shows the entire construction, fig. 3 the part that forms the „house“. Fig. 4 shows the „gate“ with servo and ranger, fig. 5 the „wall“ behind the parking spot with motion sensor and LED-string and fig. 6 the button and LED.

IN1: ranger IN2: motion sensor IN3: button

OUT1: LED OUT2: LED-string OUT3: servo

Getting started,

- opens (onboard LED is red) and close the gate to let the rover drive in (ranger and servo)
- switches the lights on on the way to the „house“ (motion sensor, LED-string)
- switches the LED-light in the „house“ on and off (button, white LED).

Programs

Rover

```
Define home1()=
Prgm
:  DispAt 1,"Coming Home - Rover"
:  Send "CONNECT RV"
:  Send "CONNECT RANGER 1 TO IN 1"
:  Send "SET COLOR 255 255 255"
:  Send "RV FORWARD SPEED 0.14 M/S"
:  d:=1
:  While d>0.25
:    Send "READ RANGER 1"
:    Get d
:  EndWhile
:  Send "RV FORWARD SPEED 0.14 M/S"
:  Wait 1.5
:  Send "RV STOP "
:  Send "SET RV.COLOR 255 0 0"
:  Wait 0.3
:  Send "SET RV.COLOR 255 255 0 BLINK 4"
:  Send "SET RV.MOTORS LEFT −150 RIGHT 0"
:  Wait 2
:  Send "RV STOP "
:  Send "SET RV.MOTORS LEFT −100 RIGHT 100"
:  Send "SET RV.COLOR 255 255 255"
:  d:=1
:  While d>0.05
:    Send "READ RV.RANGER"
:    Get d
:  EndWhile
:  Send "SET RV.COLOR 255 0 0"
:  Send "RV STOP "
:  Wait 2
:  While d<0.05
:    Send "READ RV.RANGER"
:    Get d
:  EndWhile
:  Send "SET RV.COLOR 255 255 255"
:  Send "SET RV.MOTORS LEFT −100 RIGHT 100"
:  While d>0.1
:    Send "READ RV.RANGER"
:    Get d
:  EndWhile
:  Send "RV STOP "
:  Send "SET COLOR 255 0 0"
:  Wait 0.5
:  Send "SET COLOR 0 0 0"
:EndPrgm
```

```
connecting Rover
connecting ranger external
set onboard LED to white (simulating front light)
driving rover slowly forward


scanning for gate


gate detected:
driving forward for 1.5 s

stop and LED red for 0.3 s (simulating brake light)

LED yellow blinking (simulating changing direction)
right turn


driving slowly forward
simulated front light on


reading rv.ranger


stop at d<0.05 m
brake light on



waiting in front of the gate until the gate is open

front light on
driving slowly forward


reading rv.ranger

and stop if d<0.1 m
brake light on

light off
rover is now parking
```

House:

```
Define home2()=
Prgm
:   DispAt 1,"Home2"
:   Send "CONNECT RANGER 1 TO IN 1"
:   Send "CONNECT MOTION 1 TO IN 2"
:   Send "CONNECT BUTTON 1 TO IN 3"
:   Send "CONNECT LED 1 TO OUT 1"
:   Send "CONNECT LED 2 TO OUT 2"
:   Send "CONNECT SERVO 1 TO OUT 3"
:   Send "SET SERVO 1 0"
:   a:=0
:
:   While getKey()≠"esc"
:
:     Send "READ RANGER 1"
:     Get d
:     Send "READ MOTION 1"
:     Get m
:     Send "READ BUTTON 1"
:     Get b
:
:     If d<0.2 Then
:       Wait 0.5
:       Send "SET SERVO 1 90"
:       Send "SET COLOR.RED TO ON"
:       While d<0.2
:         Send "READ RANGER 1"
:         Get d
:         Wait 0.5
:       EndWhile
:     Else
:       Send "SET SERVO 1 0"
:       Send "SET COLOR.RED TO OFF"
:   EndIf
:
:   If m=1 Then
:      Send "SET LED 2 TO ON"
:   Else
:      Send "SET LED 2 TO OFF"
:   EndIf
:
:   If b=2 and a=0 Then
:      Send "SET LED 1 TO ON"
:      a:=1
:      b:=0
:   EndIf
:   If b=2 and a=1 Then
:     Send "SET LED 1 TO OFF"
:     a:=0
:     b:=0
:   EndIf
:
:   EndWhile
:
:   Send "SET LED 1 TO OFF"
:   Send "SET LED 2 TO OFF"
:   Send "SET COLOR 0 0 0"
:EndPrgm
```

connecting the sensors

servo in home position

while-loop until „esc“ is pressed

read all sensors

if rover waits before the gate then open it

onboard RGB-LED red and on

waiting until rover drives through the open gate

servo in home position and LED off

if motion is detected then the lights outside the house are switched to on

pushing the button sets the lights in the house to on

pushing the button again sets the lights to off

end of while-loop

setting all lights to off